C3M3

Projectile Motion

We will begin by solving a ballistics problem analytically and then use Maple to solve it again. Assume that $\vec{p}(t)$ is the vector function that describes the position of the projectile at any time $t \geq 0$. We know that $\vec{p}\,'(t)$ yields the velocity of the projectile and $\vec{p}\,''(t)$ the acceleration. Further, we assume (falsely) that the only force affecting the projectile is gravity. Since gravity acts downward, our acceleration is the vector $-32\vec{j}$. We need two pieces of information about the projectile: (1) where is it at $t = 0$? and (2) what is the velocity at $t = 0$?, in order to solve this problem. What does it mean to 'solve' this problem, you may well ask? The objective is to find $\vec{p}(t)$ for $t \geq 0$. It is the basic process of starting with $\vec{p}\,''(t)$ and working 'backwards' that is being taught here.

When information about the velocity of the projectile relative to the airplane is known and the velocity of the airplane is known, how does one find the velocity of the projectile. Verbally, the answer is "The velocity of the projectile is the velocity of the airplane plus the velocity of the projectile relative to the airplane." Mathematically, it looks like

$$\vec{V}_P = \vec{V}_A + \vec{V}_{P/A}$$

Vector Example: While flying 384 feet high over level ground at 300 ft/sec, a projectile is launched from an airplane. The muzzle velocity is 200 ft/sec in the direction of $\vec{u} = .6\vec{i} + .8\vec{j}$.

(a) With the point on the ground below as reference and a gravitational constant of 32 ft/sec/sec, find a position function $\vec{p}(t)$ that determines the position of the projectile at any time t.

We note first that $\vec{u}$ is a unit vector. The velocity of the airplane is $\vec{V}_A = 300\vec{i}$. The velocity of the projectile *relative to the airplane* is $\vec{V}_{P/A} = 200\,\vec{u} = 200(.6\vec{i} + .8\vec{j}) = 120\vec{i} + 160\vec{j}$. Thus the initial velocity of the projectile is

$$\vec{p}\,'(0) = \vec{V}_P = \vec{V}_A + \vec{V}_{P/A} = 300\vec{i} + 120\vec{i} + 160\vec{j} = 420\vec{i} + 160\vec{j}$$

The initial position of the projectile is $\vec{p}(0) = 384\vec{j}$.

We begin with the vector equation $\vec{p}\,''(t) = \vec{a}(t) = -32\vec{j}$. Antidifferentiating we get $\vec{p}\,'(t) = \vec{v}(t) = -32t\vec{j} + \vec{c}$. Because $\vec{p}\,'(0) = 420\vec{i} + 160\vec{j}$, and $\vec{p}\,'(0) = \vec{c}$, we have

$$\vec{p}\,'(t) = -32t\vec{j} + 420\vec{i} + 160\vec{j} = 420\vec{i} + (160 - 32t)\vec{j}$$

Antidifferentiating again, we get $\vec{p}(t) = 420t\vec{i} + (160\,t - 16\,t^2)\vec{j} + \vec{c}_1$.
But $\vec{p}(0) = 384\,\vec{j}$, so we have our position function

$$\vec{p}(t) = 420t\vec{i} + (384 + 160t - 16t^2)\vec{j}$$

(b) When does the projectile land?

With $y(t)$ as the vertical component function, we see that we must have $y(t) = 0$.

$$\begin{aligned} y(t) = 384 + 160t - 16t^2 &= 0 \\ -16(t^2 - 10t - 24) &= 0 \\ (t - 12)(t + 2) &= 0 \\ t^* &= 12 \end{aligned}$$

(c) Where does it land?

The horizontal component function is $x(t) = 420t$. So the projectile lands $420 \cdot t^* = 420 \cdot 12 = 5040$ feet from the reference point.

(d) What is the speed at impact?

We must evaluate $\vec{v}(t^*) = \vec{v}(12) = 420\vec{i} + (160 - 32 \cdot 12)\vec{j} = 420\vec{i} - 224\vec{j}$. Because speed is the length of the velocity vector, we have an impact speed in ft/sec of

$||\vec{v}(12)|| = ||420\vec{i} - 224\vec{j}|| = \sqrt{420^2 + 224^2} = 476$

(e) When does it reach maximum height?

We must maximize $y(t)$, so set $y'(t) = 0$. This means that $160 - 32t = 0$ or $t = 5$seconds. Evaluating $y(t)$ at $t = 5$ yields a height of $384 + 160 \cdot 5 - 16 \cdot 25 = 784$ feet.

Before we begin the Maple solution it is important to understand that we cannot take a vector function such as $\langle f(t), g(t)\rangle$ in Maple and just integrate it. Note the syntax of the lines where `map` occurs. It allows the operation `int` to operate on each component. Further, when we substitute into such a function `op` allows access to each component of the vector function. If `vec1` is a vector, then `vec1[1]` refers to the first component of `vec1` and `vec1[2]` refers to the second component. The expressions `lhs, rhs` isolate the left hand side and right hand side respectively of an equation. Please take a good look at the operations in this section, because when we use them later we will assume that they are known to you.

Maple Example Use Maple to solve the projectile problem in the Vector Example.

(a) Find $\vec{p}(t)$.

```
> with(student):  with(linalg):
> acc:=vector([0,-32]):  va:=vector([300,0]):  p0:=vector([0,384]):
> u:=vector([.6,.8]):
> vpa:=evalm(200*u);
```

$$vpa := [120, 160]$$

```
> vp0:=evalm(va+vpa);
```

$$vp0 := [420, 160]$$

Set $\vec{p}\,''(t) = -32\vec{j} =$ acc.

```
> p2prime:=evalm(acc);
```

$$p2prime := [0, -32]$$

We will need vectors as constants of integration.

```
> con1:=vector([c1,c2]):  con2:=vector([c3,c4]):
> p1prime:=evalm(map(int,p2prime,t)+con1);
```

$$p1prime := [c1, -32 + c2]$$

```
> eq1:=evalm(subs(t=0,op(p1prime))=vp0);
```

$$eq1 := [c1, c2] = [420, 160]$$

```
> p1prime:=subs(c1=rhs(eq1)[1],c2=rhs(eq1)[2],op(p1prime));
```

$$p1prime := [420, -32t + 160]$$

```
> p:=evalm(map(int,p1prime,t)+con2);
```

$$p := [420t + c3, -16t^2 + 160t + c4]$$

```
> eq2:=evalm(subs(t=0,op(p))=p0);
```

$$eq2 := [c3, c4] = [0, 384]$$

```
> p:=subs(c3=rhs(eq2)[1],c4=rhs(eq2)[2],op(p));
```

$$p := [420t, -16t^2 + 160t + 384]$$

And we have found $\vec{p}(t)$.

(b) When does the projectile land? Set the second component of $\vec{p}(t)$ equal to 0.

```
> eq3:=p[2]=0;
```

$$eq3 := -16t^2 + 160t + 384 = 0$$

```
> ans:=solve(eq3,t);
```

$$ans := -2, 12$$

(c) Where does it land? Note how we access the second answer, which is the time it lands.

```
> lands:=subs(t=ans[2],op(p));
```

$$lands := [5040, 0]$$

(d) How fast was it going at impact? We must go back to $\vec{p}\,'(t)$.

```
> vel:=subs(t=ans[2],op(p1prime));
```

$$vel := [420, -224]$$

```
> speed:=norm(vel,2);
```

$$speed := 476$$

(e) How high does the projectile go? We must maximize the second component.

```
> eq4:=p1prime[2]=0;
```

$$eq4 := -32t + 160 = 0$$

```
> t1:=solve(eq4,t);
```

$$t1 := 5$$

```
> height:=subs(t=t1,p[2]);
```

$$height := 784$$

C3M3 Problem Use Maple to solve the problem below.

A plane flies in low and ascends at 600 ft/sec in the direction of $.8\vec{i} + .6\vec{j}$ where $\vec{i}$, $\vec{j}$ have the usual orientations. At the time that the plane is $12,000$ feet high, a projectile is fired with a muzzle velocity of 400 ft/sec in the direction of $\frac{\sqrt{3}}{2}\vec{i} + \frac{1}{2}\vec{j}$. Assume that the gravitational constant $g = 32\ ft/sec^2$.

(a) Find a position function $\vec{p}(t)$ for the projectile for any time $t \geq 0$. Let the point on the ground below the plane when it fires the projectile be the reference point.

(b) Find the time t^* when the projectile strikes the ground.

(c) Find the distance from the reference point to the point of impact.

(d) Find the speed of the projectile at the time of impact.
(i) in feet per second
(ii) in miles per hour. Recall that 60 miles per hour is 88 feet per second.

(e) Find the maximum height attained by the projectile and the time it occurs.